	实验五 外部中断实验（急救车与交通灯）

一.实验要求

在实验四内容的基础上增加允许急救车优先通过的要求。有急救车到达时，两向交通信号为全红，以便让急救车通过。假定急救车通过路口时间为10秒，急救车通过后，交通灯恢复中断前状态。本实验以按键为中断申请，表示有急救车通过。

二.实验目的

1.学习外部中断技术的基本使用方法。
2.学习中断处理程序的编程方法。

三.实验电路及连线

PO0-PO3接DG1-DG4,PO4-P07接 DR1-DR4。CS273 接8300H。K8接P3.2。

四.实验说明

中断服务程序的关键是：1.保护进入中断时的状态，并在退出中断之前恢复进入时的状态。2.必须在中断程序中设定是否允许中断重入，即设置EX0位。
本例中使用了INT0中断，一般中断程序进入时应保护PSW，ACC以及中断程序使用但非其专用的寄存器。本例的INT0程序保护了PSW，ACC，2等三个寄存器并且在退出前恢复了这三个寄存器。另外中断程序中涉及到关键数据的设置时应关中断，即设置时不允许重入。本例中没有涉及这种情况。

实验开始时K8应在H（高电平）端，要产生中断时先拨向L（低电平）端再拨回H端。

五.实验程序框图

六.源程序
ORG 0000H
LJMP START
ORG 0003H ;INT 0 中断入口地址
LJMP INT0
ORG 0040H
START:
MOV SP,#60H
SETB EX0 ;INT 0 中断有效
SETB IT0
SETB EA
LCALL STATUS0 ;初始状态(都是红灯)
CIRCLE: LCALL STATUS1 ;南北绿灯,东西红灯
LCALL STATUS2 ;南北绿灯闪转黄灯,东西红灯
LCALL STATUS3 ;南北红灯,东西绿灯
LCALL STATUS4 ;南北红灯,东西绿灯闪转黄灯
LJMP CIRCLE
INT0:
PUSH PSW ;保护现场
PUSH 2
PUSH ACC
MOV DPTR,#8300H
MOV A,#0FH ;南北,东西都亮红灯
MOVX @DPTR,A
MOV R2,#100 ;延时10秒
LCALL DELAY
POP ACC ;恢复现场
MOVX @DPTR,A
POP 2
POP PSW
RETI
STATUS0: ;南北红灯,东西红灯
MOV DPTR,#8300H
MOV A,#0FH
MOVX @DPTR,A
MOV R2,#10 ;延时1秒
LCALL DELAY
RET
STATUS1: ;南北绿灯,东西红灯
MOV DPTR,#8300H
MOV A,#5AH ;南北绿灯,东西红灯
MOVX @DPTR,A
MOV R2,#50 ;延时5秒
LCALL DELAY
RET
STATUS2: ;南北绿灯闪转黄灯,东西红灯
MOV DPTR,#8300H
MOV R3,#03H ;绿灯闪3次
FLASH: MOV A,#5FH
MOVX @DPTR,A
MOV R2,#03H
LCALL DELAY
MOV A,#5AH
MOVX @DPTR,A
MOV R2,#03H
LCALL DELAY
DJNZ R3,FLASH
MOV A,#0AH ;南北黄灯,东西红灯
MOVX @DPTR,A
MOV R2,#10 ;延时1秒
LCALL DELAY
RET
STATUS3: ;南北红灯,东西绿灯
MOV DPTR,#8300H
MOV A,#0A5H
MOVX @DPTR,A
MOV R2,#50 ;延时5秒
LCALL DELAY
RET
STATUS4: ;南北红灯,东西绿灯闪转黄灯
MOV DPTR,#8300H
MOV R3,#03H ;绿灯闪3次
FLASH1: MOV A,#0AFH
MOVX @DPTR,A
MOV R2,#03H
LCALL DELAY
MOV A,#0A5H
MOVX @DPTR,A
MOV R2,#03H
LCALL DELAY
DJNZ R3,FLASH1
MOV A,#05H ;南北红灯,东西黄灯
MOVX @DPTR,A
MOV R2,#10 ;延时1秒
LCALL DELAY
NOP
RET
DELAY: ;延时子程序
PUSH 2
PUSH 1
PUSH 0
DELAY1: MOV 1,#00H
DELAY2: MOV 0,#0B2H
DJNZ 0,$
DJNZ 1,DELAY2 ;延时 100 mS
DJNZ 2,DELAY1
POP 0
POP 1
POP 2
RET
END

	实验六 定时器实验

一.实验要求

由8031内部定时器1，按方式1工作，即作为16位定时器使用每0.05秒钟T1溢出中断一次。P1口的P1.0-P1.7分别接八个发光二极管。要求编写程序模拟一时序控制装置。开机后第一秒钟L1，L3亮，第二秒钟L2，L4亮，第三秒钟L5，L7亮，第四秒钟L6，L8亮，第五秒L1，L3，L5，L7亮，第六秒钟L2，L4，L6，L8亮，第七秒钟八个二极管全亮，第八秒钟全灭，以后又从头开始，L1，L3亮，然后L2，L4亮......一直循环下去。

二.实验目的

1.学习8031内部计数器的使用和编程方法。

2. 进一步掌握中断处理程序的编程方法。

三.实验电路及连线

P1.0--P1.7 接 L1--L8

四.实验说明

1.关于内部计数器的编程主要是定时常数的设置和有关控制寄存器的设置。内部计数器在单片机中主要有定时器和计数器两个功能。本实验用的是定时器，有关计数器的说明请查阅实验七。

2.内部计数器用作定时器时，是对机器周期计数。每个机器周期的长度是12个振荡器周期。因为实验系统的晶振是11.0592MHz，所以定时常数的设置可按以下方法计算：
机器周期=12÷11.0592MHz＝1.0857μS
（65536-定时常数）*1.0857μS=50mS
定时常数＝4C00H 

3.定时器的有关的寄存器有工作方式寄存器TMOD和控制寄存器TCON。TMOD用于设置定时器/计数器的工作方式0～3，并确定用于定时还是用于计数。TCON主要功能是为定时器在溢出时设定标志位，并控制定时器的运行
或停止等。

4.在例程的中断服务程序中，因为中断定时常数的设置对中断程序的运行起到关键作用，所以在置数前要先关对应的中断，置数完之后再打开相应的中断。

五.实验程序框图 

    程序框图：

六.实验程序
ORG 0000H
AJMP START
ORG 001BH ;T1中断入口地址
AJMP INT_T1
ORG 0100H
START:
MOV SP,#60H
MOV TMOD,#10H ;置T1为方式1
MOV TL1,#00H ;延时50mS的时间常数
MOV TH1,#4BH
MOV R0,#00H
MOV R1,#20
SETB TR1
SETB ET1
SETB EA ;开中断
SJMP $
INT_T1: ;T1中断服务子程序
PUSH ACC ;保护现场
PUSH PSW
PUSH DPL
PUSH DPH
CLR TR1 ;关中断
MOV TL1,#00H ;延时50mS常数
MOV TH1,#4BH
SETB TR1 ;开中断
DJNZ R1,EXIT
MOV R1,#20 ;延时一秒的常数
MOV DPTR,#DATA ;置常数表基址
MOV A,R0 ;置常数表偏移量
MOVC A,@A+DPTR ;读常数表
MOV P1,A ;送P1口显示
INC R0
ANL 00,#07H
EXIT:
POP DPH ;恢复现场
POP DPL
POP PSW
POP ACC
RETI
;LED显示常数表
DATA: DB 0FAH,0F5H,0AFH,05FH,0AAH,55H,00H,0FFH
END

	实验七 计数器实验

一.实验要求

8031内部定时计数器，按计数器模式和方式1工作，对P3.4（T0）引脚进行计数。使用8031的T1作定时器，50ms中断一次，看T0内每0.50ms来了多少脉冲，将其数值按二进制数在74LS273驱动LED灯上显示出来，5秒后再次测试。

二.实验目的

1.学习8031内部定时/计数器使用方法。

2.进一步掌握中断处理编程方法。

三.实验电路及连线

P3.4接一计数脉冲( 25模块中的149.06K频率)。 PO0~PO7接 L1~L8。 CS273接8300H。

四.实验说明

1.本实验中内部计数器起计数器的作用。外部事件计数脉冲由P3.4引入定时器T0。单片机在每个机器周期采样一次输入波形，因此单片机至少需要两个机器周期才能检测到一次跳变。这就要求被采样电平至少维持一个完整的机器周期，以保证电平在变化之前即被采样。同时这就决定了输入波形的频率不能超过机器周期频率。

2.实验时P3.4对应的连线为149.06KHZ孔，0.5ms内来了80次（50H）脉冲。

五.实验程序框图 

六.实验程序：

ORG 0000H
AJMP START
ORG 000BH ;T0中断入口地址
RETI
ORG 001BH ;T1中断入口地址
AJMP INT_T1
ORG 0040H
START:
MOV SP,#60H
MOV TMOD,#1DH ;置T1为方式1
              ;置T0为方式1,计数方式,
              ;门控选通位有效
MOV TL0,#0H ;计数器清零
MOV TH0,#0H
MOV TL1,#32H ;延时0.50mS常数
MOV TH1,#0FEH
MOV R0,#20 ;延时 5 秒常数
SETB TR0 ;启动T0
SETB ET0
SETB TR1 ;启动T1
SETB ET1
SETB EA ;开中断
SJMP $
INT_T1:
PUSH ACC
PUSH PSW
CLR TR1
MOV TL1,#32H ;延时0.50mS 常数
MOV TH1,#0FEH
MOV R1,TL0 ;保存计数值
MOV R2,TH0
MOV TL0,#00H ;清计数器
MOV TH0,#00H
SETB TR1
DJNZ R0,EXIT
MOV R0,#20 ;延时 5 秒常数
MOV DPTR,#8300H ;LED地址
MOV A,R1
MOVX @DPTR,A ;把计数值在LED上显示
EXIT:
POP PSW
POP ACC
RETI
END

	实验八 8255输入、输出实验

一.实验要求

编写程序，使用8255可编程并行口芯片，来检测八位拨动开关的状态，并控制八位发光二极管。使得八位发光二极管的亮灭变化与八位拨动开关的状态相一致。实验中用8255PB口作输入，PC口做输出。

二.实验目的

了解8255芯片结构及编程方法。

三.实验电路及连线

8255的PB0~PB7（PB口）接至八位拨动开关K1~K8。
将PC0~PC7（PC口）接至八位发光二极管L1 -- L8。CS8255接8500H。

四.实验说明

可编程通用接口芯片8255A有三个八位的并行I/O口，请用户仔细阅读有关书籍掌握其特点和各种用法。
由于8255的A1、A2脚分别接至地址线的A0、A1。所以相对应8255各口/寄存器的地址分配如下：PA口：8500H; PB口：8502H；PC口：8504H；控制寄存器地址：8506H。
本示例程序中使用PB口作输入，检测八位拨动开关的状态；使用PC口作输出，控制八个发光二极管的亮灭。

五.实验程序框图 
实验程序流程框图如下：

六.实验程序见: 

;---------------------------------------------------------
D8255 EQU 8506H ;8255 状态/命令口地址
D8255A EQU 8500H ;8255 PA 口地址
D8255B EQU 8502H ;8255 PB 口地址
D8255C EQU 8504H ;8255 PC 口地址
;---------------------------------------------------------
ORG 0000H
LJMP START
ORG 0100H
START:
LCALL DELAY ;延时
MOV DPTR,#D8255
MOV A,#82H ;置8255状态
;方式0，PA，PC口输出，PB口输入
MOVX @DPTR,A
ROTATE:
MOV DPTR,#D8255B
MOVX A,@DPTR ;读开关状态
MOV DPTR,#D8255C
MOVX @DPTR,A ;点亮对应的LED
SJMP ROTATE

DELAY: MOV R0,#0H ;延时子程序
DELAY1: MOV R1,#0H
DJNZ R1,$
DJNZ R0,DELAY1
RET
END


